Object Constraint Language Specification

version 1.1
1 September 1997
### Contents

1. **OVERVIEW** ............................................. 1
   1.1 Why OCL? .......................................................... 1
   1.2 Where to Use OCL .............................................. 2

2. **INTRODUCTION** ........................................ 2
   2.1 Legend ................................................................. 2
   2.2 Example Class Diagram ....................................... 2

3. **CONNECTION WITH THE UML METAMODEL** ................. 3
   3.1 Self ................................................................. 3
   3.2 Invariants ............................................................ 3
   3.3 Pre- and Postconditions ........................................ 4
   3.4 Guards .............................................................. 4
   3.5 General Expressions ............................................ 4

4. **BASIC VALUES AND TYPES** .................................. 4
   4.1 Types from the UML Model ................................... 5
   4.2 Enumeration Types ............................................. 5
   4.3 Type Conformance .............................................. 5
   4.4 Re-typing or Casting ........................................... 6
   4.5 Precedence Rules ................................................. 6
   4.6 Comment ............................................................ 7
   4.7 Undefined Values .............................................. 7

5. **OBJECTS AND PROPERTIES** ................................. 7
   5.1 Properties ........................................................ 7
   5.2 Properties: Attributes ......................................... 8
   5.3 Properties: Operations ........................................ 8
   5.4 Properties: Association Ends and Navigation ................. 8
   5.5 Navigation to Association Types ................................ 10
   5.6 Navigation from Association Classes .......................... 10
   5.7 Navigation through Qualified Associations ..................... 11
   5.8 Using Pathnames for Packages and Properties ................. 11
   5.9 Predefined Features on all Objects ............................ 12
   5.10 Features on Types Themselves ................................ 12
   5.11 Collections ....................................................... 12
   5.12 Collections of Collections .................................... 13
   5.13 Collection Type Hierarchy and Type Conformance Rules ....... 14
   5.14 Previous Values in Postconditions ........................... 14

6. **COLLECTION OPERATIONS** ................................ 15
   6.1 Select and Reject Operations ................................ 15
   6.2 Collect Operation ............................................... 17
   6.3 ForAll Operation ................................................. 18
   6.4 Exists Operation ................................................ 18
   6.5 Iterate Operation ............................................... 19

7. **PREDEFINED OCL TYPES** ................................. 19
   7.1 Basic Types ....................................................... 19
   7.2 Collection-Related Types ..................................... 25

8. **GRAMMAR FOR OCL** ......................................... 31
1. OVERVIEW

This document introduces and defines the Object Constraint Language (OCL), a formal language to express side effect-free constraints. Users of the Unified Modeling Language and other languages can use OCL to specify constraints and other expressions attached to their models.

OCL was used in the *UML Semantics* document to specify the well-formedness rules of the UML metamodel. Each well-formedness rule in the static semantics sections in the *UML Semantics* document contains an OCL expression, which is an invariant for the involved class. The grammar for OCL is specified at the end of this document. A parser generated from this grammar has correctly parsed all the constraints in the *UML Semantics* document, a process which improved the correctness of the specifications for OCL and UML.

1.1 Why OCL?

In object-oriented modeling a graphical model, like a class model, is not enough for a precise and unambiguous specification. There is a need to describe additional constraints about the objects in the model. Such constraints are often described in natural language. Practice has shown that this will always result in ambiguities. In order to write unambiguous constraints, so-called formal languages have been developed. The disadvantage of traditional formal languages is that they are usable to persons with a string mathematical background, but difficult for the average business or system modeler to use.

OCL has been developed to fill this gap. It is a formal language that remains easy to read and write. It has been developed as a business modeling language within the IBM Insurance division, and has its roots in the Syntropy method.

OCL is a pure expression language. Therefore, an OCL expression is guaranteed to be without side effect; it cannot change anything in the model. This means that the state of the system will never change because of an OCL expression, even though an OCL expression can be used to specify a state change, e.g. in a post-condition. All values for all objects, including all links, will not change. Whenever an OCL expression is evaluated, it simply delivers a value.

OCL is not a programming language, so it is not possible to write program logic or flow control in OCL. You cannot invoke processes or activate non-query operations within OCL. Because OCL is a modeling language in the first place, not everything in it is promised to be directly executable.

OCL is a typed language, so each OCL expression has a type. In a correct OCL expression all types used must be type conformant. For example, you cannot compare an Integer with a String. Types within OCL can be any kind of Classifier within UML.

As a modeling language, all implementation issues are out of scope and cannot be expressed in OCL. Each OCL expression is conceptually atomic. The state of the objects in the system cannot change during evaluation.
1.2 WHERE TO USE OCL

OCL can be used for a number of different purposes:

- To specify invariants on classes and types in the class model.
- To specify type invariant for Stereotypes.
- To describe pre- and post conditions on Operations and Methods
- To describe Guards
- As a navigation language
- To specify constraints on operations:
  \[ \text{operation} = \text{expression} \]

Where \textit{operation} is the name of the operation and \textit{expression} the constraint. Because operations may have parameters, the constraint may also have one or more parameters as in one of the following:

\[ \text{operation}(a, b) = \text{expression} \]
\[ \text{operation}(a : \text{Type1}, b : \text{Type2}) = \text{expression} \]

The parameters of the operation, in this example \textit{a} and \textit{b}, can be used in the expression at the right-hand side of the equals sign. Operations can also be described by a recursive expression. It is the modeler’s task to make sure that the recursion is well defined. An operation constraint can also be read as a definition of the operation, where the right-hand side of the equals sign determines the value the operation will return.

Within the \textit{UML Semantics} document, OCL is used in the well-formedness rules as invariants on the meta-classes in the abstract syntax. At several places it is also used to define ‘additional’ operations, which are used in the well-formedness rules.

2. INTRODUCTION

2.1 LEGEND

Text written in the courier typeface like below is an OCL expression:

\[ 'This is an OCL expression' \]

The underlined word before an OCL expression determines the context for the expression:

\[ \text{TypeName} 'this is an OCL expression in the context of TypeName' \]

Keywords of OCL are written in boldface within the OCL expression in this document. The boldface has no formal meaning, but is used to make the expressions more readable in this document. OCL expressions are written using only ASCII characters.

Words in \textit{Italics} within the main text of the paragraphs refer to parts of OCL expressions.

2.2 EXAMPLE CLASS DIAGRAM

The diagram below is used in the examples in this document.
3. CONNECTION WITH THE UML METAMODEL

3.1 SELF

Each OCL expression is written in the context of an instance of a specific type. In an OCL expression the name self is used to refer to the contextual instance.

3.2 INVARIANTS

The OCL expression can be part of an Invariant, which is a Constraint stereotyped with «invariant». When the Invariant is associated with a Classifier, this is called the type in this document. The expression then is an invariant of the type and must be true for all instances of that type at any time. If the context is Company, then self refers to an instance of Company. In the expression

```
self.numberOfEmployees
```

self is an instance of type Company. We can see the self as the object from where we start the expression.

In this document, the type of the contextual instance of an OCL expression, which is part of an Invariant, is written with the name of the type underlined as follows:

```
Company

self.numberOfEmployees
```

In most cases, self can be left out, because the context is clear, as in the above examples.

As an alternative for self, a different name can be defined playing the part of self:
3.3 Pre- and Postconditions

The OCL expression can be part of a Precondition or Postcondition, which are Constraints stereotyped with respectively «precondition» and «postcondition». The Precondition or Postcondition on Operation or Method. In this case, the expression is a pre- or postcondition on the Operation or Method. The contextual instance self then is of the type which owns the operation as a feature. The notation used in this document is to underline the type and operation declaration, and put labels ‘pre:’ and ‘post:’ before Preconditions and Postconditions

```
Typename::operationName(parameter1 : Typel, ... ) : ReturnType
pre :  parameterl > ...
post:  result = ...
```

The name self can be used in the expression referring to the object on which the operation was called, and the name result is the name of the returned object, if there is any. The names of the parameters (parameter1, ) can also be used in the OCL expression. In the example diagram we can write:

```
Person::income(d : Date) : Integer
pre:  result = ...some function of self and parameter1 ...
```

3.4 Guards

The OCL expression can be part of a Guard. In this case self refers to the enclosing Classifier. No examples of guards are given in this document.

3.5 General Expressions

Any OCL expression can be used as the value for an attribute of the UML class Expression or one of its subtypes. In this case, the semantics document describes the meaning of the expression.

4. Basic Values and Types

In OCL, a number of basic types are predefined and available to the modeler at all time. These predefined value types are independent of any object model and part of the definition of OCL.

The most basic value in OCL is a value of one of the basic types. Some basic types used in the examples in this document, with corresponding examples of their values, are:

<table>
<thead>
<tr>
<th>type</th>
<th>values</th>
</tr>
</thead>
<tbody>
<tr>
<td>Boolean</td>
<td>true, false</td>
</tr>
<tr>
<td>Integer</td>
<td>1, 2, 34, 26524, ...</td>
</tr>
<tr>
<td>Real</td>
<td>1.5, 3.14, ...</td>
</tr>
<tr>
<td>String</td>
<td>'To be or not to be...'</td>
</tr>
</tbody>
</table>
OCL defines a number of operations on the predefined types. The next table gives some examples of the operations on the predefined types. In section 7 the complete list of all operations is given.

<table>
<thead>
<tr>
<th>type</th>
<th>operations</th>
</tr>
</thead>
<tbody>
<tr>
<td>Integer</td>
<td>*, +, -, /, abs</td>
</tr>
<tr>
<td>Real</td>
<td>*, +, -, /, floor</td>
</tr>
<tr>
<td>Boolean</td>
<td>and, or, xor, not, implies, if-then-else</td>
</tr>
<tr>
<td>String</td>
<td>toUpper, concat</td>
</tr>
</tbody>
</table>

At the end of this document the complete list of operations provided for each type is described. Collection, Set, Bag and Sequence are basic types as well; their specifics will be described in the upcoming sections.

4.1 TYPES FROM THE UML MODEL

Each OCL expression is written in the context of a UML model, a number of types/classes their features and associations and their generalizations. All types/classes from the UML model are types in OCL that is attached to the model.

4.2 ENUMERATION TYPES

As shown in the example diagram, new enumeration types can be defined in a model by using:

```plaintext
e num{ value1, value2, value3 }
```

The values of the enumeration (value1, ...) can be used within expressions.

As there might be a name conflict with attribute names being equal to enumeration values, the usage of an enumeration value is syntactically expressed with an additional # symbol in front of the value:

```plaintext
#value1
```

The type of an enumeration attribute is Enumeration, with restrictions on the values for the attribute.

4.3 TYPE CONFORMANCE

OCL is a typed language and the basic value types are organized in a type hierarchy. This hierarchy determines conformance of the different types to each other. You cannot, for example, compare an Integer with a Boolean or a String.

An OCL expression in which all the types conform is a valid expression. An OCL expression in which the types don’t conform is an invalid expression. It contains a type conformance error. A type `type1` conforms to a type `type2` when an instance of `type1` can be substituted at each place where an instance of `type2` is expected. The type conformance rules for types in the class diagrams are simple.

- Each type conforms to its supertype
- Type conformance is transitive: if `type1` conforms to `type2`, and `type2` conforms to `type3`, then `type1` conforms to `type3`. 
The effect of this is that a type conforms to its supertype, and all the supertypes above.

The type conformance rules for the value types are:

<table>
<thead>
<tr>
<th>Type</th>
<th>Conforms to/ Is subtype of</th>
</tr>
</thead>
<tbody>
<tr>
<td>Set</td>
<td>Collection</td>
</tr>
<tr>
<td>Sequence</td>
<td>Collection</td>
</tr>
<tr>
<td>Bag</td>
<td>Collection</td>
</tr>
<tr>
<td>Integer</td>
<td>Real</td>
</tr>
</tbody>
</table>

The conformance relation between the collection types only holds if they are collections of element types that conform to each other. See 5.13 for the complete conformance rules for collections.

In the next table some examples of valid and invalid expressions are shown:

<table>
<thead>
<tr>
<th>OCL expression</th>
<th>valid?</th>
<th>error</th>
</tr>
</thead>
<tbody>
<tr>
<td>1 + 2 * 34</td>
<td>yes</td>
<td></td>
</tr>
<tr>
<td>1 + ‘motorcycle’</td>
<td>no</td>
<td>type Integer does not conform to type String</td>
</tr>
<tr>
<td>23 * false</td>
<td>no</td>
<td>type Integer does not conform to Boolean</td>
</tr>
<tr>
<td>12 + 13.5</td>
<td>yes</td>
<td></td>
</tr>
</tbody>
</table>

### 4.4 Re-typing or Casting

In some circumstances it is desirable to use a property of an object that is defined on a subtype of the current known type of the object. Because the property is not defined on the current known type this results in a type conformance error.

When it is certain that the actual type of the object is the subtype, the object can be re-typed using the operation `oclAsType(OclType)`. This operation results in the same object, but the known type is the argument `OclType`. When there is an object `object` of type `Type1` and `Type2` is another type, it is allowed to write:

```ocl
object.oclAsType(Type2)  --- evaluates to object with type Type2
```

An object can only be re-typed to one of its subtype; therefore, in the example, `Type2` must be a subtype of `Type1`.

If the actual type of the object is not equal to the type to which it is re-typed, the expression is undefined (see 4.7).

### 4.5 Precedence Rules

The precedence order for the operations in OCL is:

- dot and arrow operations have highest precedence
- unary ‘not’ and unary minus ‘-’
- ‘*’ and ‘/’
• ‘+’ and binary ‘-’
• ‘and’, ‘or’ and ‘xor’
• ‘implies’
• ‘if-then-else-endif’
• ‘<’, ‘>’, ‘<=>’, ‘=>’ and ‘=’

Parenthesis ‘(’ and ‘)’ can be used to change precedence.

4.6 COMMENT

Comments in OCL are written after two dashes. Everything after the two dashes up to and including the end of line is comment. For example:

```plaintext
-- this is a comment
```

4.7 UNDEFINED VALUES

Whenever an OCL expression is being evaluated, there is a possibility that one or more of the queries in the expression are undefined. If this is the case, then the complete expression will be undefined.

There are two exceptions to this for the boolean operators:

- True OR-ed with anything is True
- False AND-ed with anything is False

The above two rules are valid irrespective of the order of the arguments and the above rules are valid whether or not the value of the other sub-expression is known.

5. OBJECTS AND PROPERTIES

OCL expressions can refer to types, classes, interfaces, associations (acting as types) and datatypes. Also all attributes, association-ends, methods and operations without side-effects that are defined on these types etc. can be used. In a class model, an operation or method is defined to be side effect free if the isQuery attribute of the operations is true. For the purpose of this document, we will refer to attributes, association-ends, and side-effect-free methods and operations as being properties. A property therefore is one of:

- an Attribute,
- an AssociationEnd,
- an Operation with isQuery being true,
- a Method with isQuery being true

5.1 PROPERTIES

The value of a property on an object that is defined in a class diagram is specified by a dot followed by the name of the property:

```plaintext
AType
  self.property
```
If *self* is a reference to an object, then *self.property* is the value of the *property* property on *self*.

### 5.2 Properties: Attributes

For example, the age of a Person is written as

```mermaid
diagrammer
Person
  self.age
```

The value of this expression is the value of the *age* attribute on the Person *self*. The type of this expression is the type of the attribute *age*, which is the basic type Integer.

With attributes, and the operations defined on the basic value types we can express calculations etc. over the class model. For example, a business rule might be “the age of a Person is always greater or equal to zero.” This can be stated as the invariant:

```mermaid
Person
  self.age >= 0
```

### 5.3 Properties: Operations

Operations may have parameters. For example, in the example diagram shown earlier, a Person object has an income expressed as a function of the date. This operation would be accessed as follows, for a Person *aPerson* and a date *aDate*:

```mermaid
aPerson
  .income(aDate)
```

The operation itself could be defined by a postcondition constraint. This is a constraint that is stereotyped as «postcondition». The object that is returned by the operation can be referred to by `result`. It takes the following form:

```mermaid
Person::income (d: Date) : Integer
  post: result = -- some function of d and other properties of person
```

The right-hand-side of this definition may refer to the operation being defined; i.e. the definition may be recursive, as long as the recursion is well defined. The type of *result* is the return type of the operation, which is Integer in the above example.

To refer to an operation or a method that doesn’t take a parameter, parenthesis with an empty argument list are used:

```mermaid
Company
  self.stockPrice()
```

### 5.4 Properties: Association Ends and Navigation

Starting from a specific object, we can navigate an association on the class diagram to refer to other objects and their properties. To do so, we navigate the association by using the opposite association-end:

```mermaid
object.rolename
```

The value of this expression is the set of objects on the other side of the *rolename* association. If the multiplicity of the association-end has a maximum of one (“0..1” or “1”), then the value of this expression is an object. In the example class diagram, when we start in the context of a Company (i.e. *self* is an instance of Company), we can write:

```mermaid
Company
```
The evaluation of the first expression will result in an object of type Person, because the multiplicity of the association is one. The evaluation of the second expression will result in a Set of Persons. By default, navigation will result in a Set. When the association on the Class Diagram is adorned with {ordered}, the navigation results in a Sequence.

Collections, like Sets, Bags and Sequences, are predefined types in OCL. They have a large number of predefined operations on them. A property of the collection itself is accessed by using an arrow ‘->’ followed by the name of the property. The following example is in the context of a person:

```
Person
self.employer->size
```

This applies the `size` property on the Set `self.employer`, which results in the number of employers of the Person `self`.

```
Person
self.employer->isEmpty
```

This applies the `isEmpty` property on the Set `self.employer`. This evaluates to true if the set of employers is empty, and false otherwise.

### 5.4.1 Missing Rolenames

Whenever a rolename is missing at one of the ends of an association, the name of the type at the association end, starting with a lowercase character is used as the rolename. If this results in an ambiguity the rolename is mandatory. This is the case with unnamed rolenames in on reflexive associations. If the rolename is ambiguous, then it cannot be used in OCL.

### 5.4.2 Navigation over Associations with Multiplicity Zero or One

Because the multiplicity of the role `manager` is one, `self.manager` is an object of type Person. Such a single object can be used as a Set as well. It then behaves as if it is a Set containing the single object. The usage as a set is done through the arrow, followed by a property of Set. This is shown in the following example:

```
Company
self.manager->size -- 'self.manager' is used as Set, because the
-- arrow
-- is used to access the 'size' property on Set
-- This expresin result in 1

self.manager->foo  -- 'self.manager' is used as Set, because the
-- arrow is used to access the 'foo' property on
-- Set. This expresion is incorrect, since 'foo'
-- is not a defined property of Set.

self.manager.age   -- 'self.manager' is used as Person, because the
-- dot
-- is used to access the 'age' property of Person
```

In the case of an optional (0..1 multiplicity) association, this is especially useful to check whether there is an object or not when navigating the association. In the example we can write:
5.4.3 Combining Properties

Properties can be combined to make more complicated expressions. An important rule is that an OCL expression always evaluates to a specific object of a specific type. Upon this result, one can always apply another property. Therefore, each OCL expression can be read and evaluated left-to-right.

Following are some invariants that use combined properties on the example class diagram:

1. Married people are of age >= 18
   \[ \text{self.wife->notEmpty implies self.wife.age} \geq 18 \text{ and self.husband->notEmpty implies self.husband.age} \geq 18 \]

2. A company has at most 50 employees
   \[ \text{self.employee->size} \leq 50 \]

3. A marriage is between a female (wife) and male (husband)
   \[ \text{self.wife.sex} = \text{#female and self.husband.sex} = \text{#male} \]

4. A person can not both have a wife and a husband
   \[ \text{not ((self.wife->size} = 1) \text{ and (self.husband->size} = 1)) \]

5.5 Navigation to Association Types

To specify navigation to association classes (Job and Marriage in the example), OCL uses a dot and the name of the association class starting with a lowercase character:

\[ \text{Person self.job} \]

This evaluates to a Set of all the jobs a person has with the companies that are his/her employer. In the case of an association class there is no explicit rolename in the class diagram. The name \text{job} used in this navigation is the name of the association class starting with a lowercase character, similar to the way described in the section “Missing Rolenames” above.

5.6 Navigation from Association Classes

We can navigate from the association class itself to the objects that participate in the association. This is done using the dot-notation and the role-names at the association-ends.

\[ \text{Job self.employer self.employee} \]

Navigation from an association class to one of the objects on the association will always deliver exactly one object. This is a result of the definition of AssociationClass. Therefore the result of this navigation is exactly one object, although it can be used as a Set using the arrow (\text{\rightarrow}).
5.7 Navigation Through Qualified Associations

Qualified associations use one or more qualifier attributes to select the objects at the other end of the association. To navigate them, we can add the values for the qualifiers to the navigation. This is done using square brackets, following the role-name. It is permissible to leave out the qualifier values, in which case the result will be all objects at the other end of the association.

```
Bank
  self.customer          -- results in a Set(Person) containing all customers of the Bank
  self.customer[8764423] -- results in one Person, having account number 8764423
```

If there is more than one qualifier attribute, the values are separated by commas. It is not permissible to partially specify the qualifier attribute values.

5.8 Using Pathnames for Packages and Properties

Within UML, different types are organized in packages. OCL provides a way of explicitly referring to types in other packages by using a package-pathname prefix. The syntax is a package name, followed by a double colon:

```
Packagename::Typename
```

This usage of pathnames is transitive and can also be used for packages within packages:

```
Packagename1::Packagename2::Typename
```

Whenever properties are redefined within a type, the property of the supertypes can be accessed using the same path syntax. Whenever we have a class B, as a subtype of class A, and a property \( p1 \) of both A and B. We can write:

```
B
  self.A::p1  -- accesses the \( p1 \) property defined in A
  self.B::p1  -- accesses the \( p1 \) property defined in B
```

The following shows an example where such a pathname is needed:

```
....
ModelElement
  Note
    value: Uninterpreted
  Dependency
    source
    target
Dependency
  self.source
```

In this model fragment there is an ambiguity with the OCL expression on Dependency:

```
Dependency
  self.source
```
This can either mean normal association navigation, which is inherited from ModelElement. It might also mean navigation through the dotted line as an association class. Both possible navigations use the same role-name, so this is always ambiguous. Using the pathname we can distinguish between them with:

```
Dependency
   self.Dependency::source
   self.ModelElement::source
```

### 5.9 Predefined Features on all Objects

There are several features that apply to all objects, and are predefined in OCL. These are:

- `oclType` : `OclType`
- `oclIsTypeOf(t : OclType) : boolean`
- `oclIsKindOf(t : OclType) : boolean`

The feature `oclType` results in the type of an object. For example, the expression

```
Person
   self.oclType
```

results in `Person`. The type of this is `OclType`, a predefined type within the OCL language. (NB: not `Person`, which is the type of `self`)

The operation `isTypeOf` results in true if the type of `self` and `t` are the same. For example:

```
Person
   self.oclIsTypeOf( Person )  -- is true
   self.oclIsTypeOf( Company ) -- is false
```

The above feature deals with the direct type of an object. The `oclIsKindOf` feature determines whether `t` is either the direct type or one of the supertypes of an object.

### 5.10 Features on Types Themselves

All properties discussed until now in OCL are properties on instances of classes. The types are either predefined in OCL or defined in the class model. In OCL, it is also possible to use features defined on the types/classes themselves. These are, for example, the class-scoped features defined in the class model. Furthermore, several features are predefined on each type.

The most important predefined feature on each type is `allInstances`, which results in the Set of all instances of the type. If we want to make sure that all instances of `Person` have unique names we can write:

```
Person.allInstances->forAll(p1, p2 | p1 <> p2 implies p1.name <> p2.name)
```

The `Person.allInstances` is the set of all persons and is of type `Set(Person)`.

### 5.11 Collections

Navigation will most often result in a collection; therefore, the collection types play an important role in OCL expressions.

The type `Collection` is predefined in OCL. The Collection type defines a large number of predefined operations to enable the OCL expression author (the modeler) to manipulate
collections. Consistent with the definition of OCL as an expression language, collection operations never change collections; isQuery is always true. They may result in a collection, but rather than changing the original collection they project the result into a new one.

Collection is an abstract type, with the concrete collection types as its subtypes. OCL distinguishes three different collection types: Set, Sequence, and Bag. A Set is the mathematical set. It does not contain duplicate elements. A Bag is like a set, which may contain duplicates, i.e. the same element may be in a bag twice or more. A Sequence is like a Bag in which the elements are ordered. Both Bags and Sets have no order defined on them. Sets, Sequences, and Bags can be specified by a literal in OCL. Curly brackets surround the elements of the collection; elements in the collection are written within, separated by commas. The type of the collection is written before the curly brackets:

Set `{ 1 , 2 , 5 , 88 }`
Set `{ 'apple' , 'orange' , 'strawberry' }`

A Sequence:

Sequence `{ 1 , 3 , 45 , 2 , 3 }`
Sequence `{ 'ape' , 'nut' }`

A Bag:

Bag `{ 1 , 3 , 4 , 3 , 5 }`

Because of the usefulness of a Sequence of consecutive Integers, there is a separate literal to create them. The elements inside the curly brackets can be replaced by an interval specification, which consists of two expression of type Integer, `Int-expr1` and `Int-expr2`, separated by `. ..`. This denotes all the Integers between the values of `Int-expr1` and `Int-expr2`, including the values of `Int-expr1` and `Int-expr2` themselves:

Sequence `{ 1 .. (6 + 4) }`
Sequence `{ 1 .. 10 }`
-- are both identical to
Sequence `{ 1 , 2 , 3 , 4 , 5 , 6 , 7 , 8 , 9 , 10 }`

The complete list of Collection operations is described at the end of this document.

Collections can be specified by a literal, as described above. The only other way to get a collection is by navigation. To be more precise, the only way of getting a Set, Sequence, or Bag is:

- a literal, this will result in a Set, Sequence or Bag:
  
  Set `{ 1 , 2 , 3 , 5 , 7 , 11 , 13 , 17 }`
  Sequence `{ 1 , 2 , 3 , 5 , 7 , 11 , 13 , 17 }`
  Bag `{ 1 , 2 , 3 , 2 , 1 }`

- a navigation starting from a single object can result in a collection e.g.:
  
  Company
  self.employee

- operations on collections may result in new collections. E.g.:
  
  collection1->union(collection2)

**5.12 COLLECTIONS OF COLLECTIONS**

Within OCL, all Collections of Collections are automatically flattened. Therefore the following two expressions have the same value:
5.13 Collection Type Hierarchy and Type Conformance Rules

In addition to the type conformance rules in section 4.3 the following rules hold for all types, including the collection types:

- Every type Collection (X) is a subtype of OclAny. The types Set (X), Bag (X) and Sequence (X) are all subtypes of Collection (X).

Type conformance rules are as follows for the collection types:

- \( \text{Type1} \) conforms to \( \text{Type2} \) when they are identical (standard rule for all types).
- \( \text{Type1} \) conforms to \( \text{Type2} \) when it is a subtype of \( \text{Type2} \) (standard rule for all types).
- \( \text{Collection}(\text{Type1}) \) conforms to \( \text{Collection}(\text{Type2}) \), when \( \text{Type1} \) conforms to \( \text{Type2} \).
- Type conformance is transitive: if \( \text{Type1} \) conforms to \( \text{Type2} \), and \( \text{Type2} \) conforms to \( \text{Type3} \), then \( \text{Type1} \) conforms to \( \text{Type3} \) (standard rule for all types).

For example, if Bicycle and Car are two separate subtypes of Transport:

\[
\begin{align*}
\text{Set}(\text{Bicycle}) & \text{ conforms to } \text{Set}(\text{Transport}) \\
\text{Set}(\text{Bicycle}) & \text{ conforms to } \text{Collection}(\text{Bicycle}) \\
\text{Set}(\text{Bicycle}) & \text{ conforms to } \text{Collection}(\text{Transport})
\end{align*}
\]

Note that Set(Bicycle) does not conform to Bag(Bicycle), nor the other way around. They are both subtypes of Collection(Bicycle) at the same level in the hierarchy.

5.14 Previous Values in Postconditions

As stated in section 3.3, OCL can be used to specify pre- and post-conditions on Operations and Methods in UML. In a postcondition, the expression can refer to two sets of values for each property of an object:

- the value of a property at the start of the operation or method
- the value of a property upon completion of the operation or method

The value of a property in a postcondition is the values upon completion of the operation. To refer to the value of a property at the start of the operation, one has to postfix the property-name with the commercial at sign ‘@’, followed by the keyword ‘pre’:

\[
\text{Person::birthdayHappens()}
\]

\[
\text{post: age = age@pre + 1}
\]

The property \( \text{age} \) refers to the property of the instance of Person on which executes the operation. The property \( \text{age}@\text{pre} \) refers to the value of the property \( \text{age} \) of the Person that executes the operation, at the start of the operation.

If the property has parameters, the ‘@pre’ is postfixed to the propertyname, before the parameters.

\[
\text{Company::hireEmployee(p : Person)}
\]

\[
\text{post: employees = employees@pre->including(p) and}
\]
stockprice() = stockprice@pre() + 10

The above operation can also be specified by a post and pre condition together

\begin{verbatim}
Company::hireEmployee(p : Person)
    pre  : not employee->includes(p)
    post: employees->includes(p) and
          stockprice() = stockprice@pre() + 10
\end{verbatim}

When the pre-value of a property is takes and this evaluates to an object, all further properties that are accessed of this object are the new values (upon completion of the operation) of this object. So:

\begin{verbatim}
a.b@pre.c       -- takes the old value of property b of a, say x
                -- and then the new value of c of x.
a.b@pre.c@pre   -- takes the old value of property b of a, say x
                -- and then the old value of c of x.
\end{verbatim}

The ‘@pre’ postfix is only allowed in OCL expressions that are part of a Postcondition. Asking for a current property of an object that has been destroyed during execution of the operation results in Undefined. Also, referring to the previous value of an object that has been created during execution of the operation results in Undefined.

6. COLLECTION OPERATIONS

OCL defines many operations on the collection types. These operations are specifically meant to enable a flexible and powerful way of projecting new collections from existing ones. The different constructs are described in the following sections.

6.1 SELECT AND REJECT OPERATIONS

Sometimes an expression using operations and navigations delivers a collection, while we are interested only in a special subset of the collection. OCL has special constructs to specify a selection from a specific collection. These are the select and reject operations. The select specifies a subset of a collection. A select is an operation on a collection and is specified using the arrow-syntax:

\begin{verbatim}
collection->select( ... )
\end{verbatim}

The parameter of select has a special syntax that enables one to specify which elements of the collection we want to select. There are three different forms, of which the simplest one is:

\begin{verbatim}
collection->select( boolean-expression )
\end{verbatim}

This results in a collection that contains all the elements from collection for which the boolean-expression evaluates to true. To find the result of this expression, for each element in collection the expression boolean-expression is evaluated. If this evaluates to true, the element is included in the result collection, otherwise not. As an example, the next OCL expression specifies all the employees older than 50 years:

\begin{verbatim}
Company
    self.employee->select(age > 50)
\end{verbatim}

The self.employee is of type Set(Person). The select takes each person from self.employee and evaluates age > 50 for this person. If this results in true, than the person is in the result Set.
As shown in the previous example, the context for the expression in the select argument is the element of the collection on which the select is invoked. Thus the `age` property is taken in the context of a person.

In the above example, it is impossible to refer explicitly to the persons themselves; you can only refer to properties of them. To enable to refer to the persons themselves there is a more general syntax for the select expression:

```
Collection->select( v | boolean-expression-with-v )
```

The variable `v` is called the iterator. When the select is evaluated, `v` iterates over the `collection` and the `boolean-expression-with-v` is evaluated for each `v`. The `v` is a reference to the object from the collection and can be used to refer to the objects themselves from the `collection`. The two examples below are identical:

```java
Company
self.employee->select(age > 50)
```

```java
Company
self.employee->select(p | p.age > 50)
```

The result of the complete select is the collection of persons `p` for which the `p.age > 50` evaluates to True. This amounts to a subset of `self.employee`.

As a final extension to the select syntax the expected type of the variable `v` can be given. The select now is written as:

```
Collection->select( v : Type | boolean-expression-with-v )
```

The meaning of this is that the objects in `collection` must be of type `Type`. The next example is identical to the previous examples:

```java
Company
self.employee.select(p : Person | p.age > 50)
```

The complete select syntax now looks like one of:

```
collection->select( v : Type | boolean-expression-with-v )
collection->select( v | boolean-expression-with-v )
collection->select( boolean-expression )
```

The `Reject` operation is identical to the select operation, but with reject we get the subset of all the elements of the collection for which the expression evaluates to **False**. The reject syntax is identical to the select syntax:

```
Collection->reject( v : Type | boolean-expression-with-v )
Collection->reject( v | boolean-expression-with-v )
Collection->reject( boolean-expression )
```

As an example, specify all the employees who are **not** married:

```java
Company
self.employee->reject( isMarried )
```

The reject operation is available in OCL for convenience, because each reject can be restated as a select with the negated expression. Therefore the following two expressions are identical:

```
Collection->reject( v : Type | boolean-expression-with-v )
collection->select( v : Type | not (boolean-expression-with-v) )
```
6.2 **COLLECT OPERATION**

As shown in the previous section, the select and reject operation always result in a sub-collection of the original collection. When we want to specify a collection which is derived from some other collection, but which contains different objects from the original collection (i.e. it is not a sub-collection), we can use a collect operation. The collect operation uses the same syntax as the select and reject and is written as one of:

```
    collection->collect( v : Type | expression-with-v )
    collection->collect( v | expression-with-v )
    collection->collect( expression )
```

The value of the reject operation is the collection of the results of all the evaluations of `expression-with-v`.

An example: specify the collection of `birthDates` for all employees in the context of a company. This can be written as one of:

```
    Company
    self.employee->collect( birthDate )
    self.employee->collect( person | person.birthDate )
    self.employee->collect( person : Person | person.birthDate )
```

An important issue here is that the resulting collection is not a Set, but a Bag. When more than one employee has the same value for `birthDate`, this value will be an element of the resulting Bag more than once. The Bag resulting from the `collect` operation always has the same size as the original collection.

It is possible to make a Set from the Bag, by using the `asSet` property on the Bag. The following expression results in the Set of different `birthDates` from all employees of a Company:

```
    Company
    self.employee->collect( birthDate )->asSet
```

### 6.2.1 Shorthand for Collect

Because navigation through many objects is very common, there is a shorthand notation for the `collect` that makes the OCL expressions more readable. Instead of

```
    self.employee->collect(birthdate)
```

we can also write:

```
    self.employee.birthdate
```

In general, when we apply a property to a Collection of Objects, then it will automatically be interpreted as a `collect` over the members of the Collection with the specified property.

Therefore, for any `propertyname` that is defined as a property on the objects in a collection, the following two expressions are identical:

```
    collection.propertyname
    collection->collect(propertyname)
```

and so are those if the property is parameterized.

```
    collection.propertyname(par1, par2, ...)
    collection->collect(propertyname(par1, par2, ...)
```
6.3 **forAll Operation**

Many times a constraint is needed on all elements of a collection. The forAll operation in OCL allows specifying a Boolean expression, which must hold for all objects in a collection:

```
collection->forAll( v : Type | boolean-expression-with-v )
collection->forAll( v | boolean-expression-with-v )
collection->forAll( boolean-expression )
```

This forAll expression results in a Boolean. The result is true if the `boolean-expression-with-v` is true for all elements of `collection`. If the `boolean-expression-with-v` is false for one or more `v` in `collection`, then the complete expression evaluates to false. For example, in the context of a company:

```
Company
self.employee->forAll( forename = 'Jack' )

self.employee->forAll( p | p.forename = 'Jack' )

self.employee->forAll( Person p | p.forename = 'Jack' )
```

These expressions evaluate to true if the forename feature of each employee equals to ‘Jack’.

The forAll operation has an extended variant in which more than one iterator is used. Both iterators will iterate over the complete collection. Effectively this is a forAll on the Cartesian product of the collection with itself.

```
Company
self.employee->forAll( e1, e2 | e1 <> e2 implies e1.forename <> e2.forename )

self.employee->forAll( Person e1, e2 | e1 <> e2 implies e1.forename <> e2.forename )
```

This expression evaluates to true if the forenames of all employees are different.

It is semantically equivalent to:

```
Company
self.employee->forAll( e1 | self.employee->forAll( e2 | e1 <> e2 implies e1.forename <> e2.forename ))
```

6.4 **exists Operation**

Many times one needs to know whether there is at least one element in a collection for which a constraint holds. The exists operation in OCL allows to specify a Boolean expression which must hold for at least one object in a collection:

```
collection->exists( v : Type | boolean-expression-with-v )
collection->exists( v | boolean-expression-with-v )
collection->exists( boolean-expression )
```

This forAll operation results in a Boolean. The result is true if the `boolean-expression-with-v` is true for at least one element of `collection`. If the `boolean-expression-with-v` is false for all `v` in `collection`, then the complete expression evaluates to false. For example, in the context of a company:

```
Company
self.employee->exists( forename = 'Jack' )
self.employee->exists( p | p.forename = 'Jack' )
self.employee->exists( Person p | p.forename = 'Jack' )
```

These expressions evaluate to true if the forename feature of at least one employee equals to ‘Jack’.
6.5 **Iterate Operation**

The iterate operation is slightly more complicated, but is a very generic. The operations reject, select, forAll, exists, collect, elect can all be described in terms of iterate.

An accumulation build one value by iterating over a collection.

```
collection->iterate( elem : Type; acc : Type = <expression> | expression-with-elem-and-acc )
```

The variable `elem` is the iterator, as in the definition of select, forAll, etc. The variable `acc` is the accumulator. The accumulator gets an initial value `<expression>`.

When the iterate is evaluated, `elem` iterates over the `collection` and the `expression-with-elem-and-acc` is evaluated for each `elem`. After each evaluation of `expression-with-elem-and-acc`, its value is assigned to `acc`. In this way the value of `acc` is build up during the iteration of the collection.

The collect operation described in terms of iterate will look like:

```
collection->collect(x : T | x.property)
```

-- is identical to:

```
collection->iterate(x : T; acc : T2 = Bag{} | acc->including(x.property))
```

Or written in Java-like pseudocode the result of the iterate can be calculated as:

```
iterate(elem : T; acc : T2 = value)
{
    acc = value;

    for(Enumeration e = collection.elements() ; e.hasMoreElements(); ){
        elem = e.nextElement();
        acc  = <expression-with-elem-and-acc>
    }
}
```

7. **Predefined OCL Types**

This section contains all standard types defined within OCL, including all the features defined on those types. Its signature and a description of its semantics define each feature. Within the description the name ‘result’ is used to refer to the value that results from evaluating the feature. In several places post conditions are used to describe properties of the result. When there is more than one postcondition, all postconditions must be true.

7.1 **Basic Types**

The basic types used are Integer, Real, String and Boolean. They are supplemented with OclExpression, OclType and OclAny.

7.1.1 **OclType**

All types defined in an UML model, or pre-defined within OCL have a type. This type is an instance of the OCL type called OclType. Access to this type allows the modeler access to the meta-level of the model. This can be useful for advanced modelers.
Features of OclType, the instance of OclType is called type.

- **type.name** : String
  The name of type.

- **type.attributes** : Set(String)
  The set of names of the attributes of type, as they are defined in the model.

- **type.associationEnds** : Set(String)
  The set of names of the navigable associationEnds of type, as they are defined in the model.

- **type.operations** : Set(String)
  The set of names of the operations of type, as they are defined in the model.

- **type.supertypes** : Set(OclType)
  The set of all direct supertypes of type.
  \[\text{post: type.allSupertypes->includesAll(result)}\]

- **type.allSupertypes** : Set(OclType)
  The transitive closure of the set of all supertypes of type.

- **type.allInstances** : Set(type)
  The set of all instances of type and all its subtypes.

### 7.1.2 OclAny

Within the OCL context, the type OclAny is the supertype of all types in the model. Features on OclAny are available on each object in all OCL expressions.

All classes in a UML model inherit all features defined on OclAny. To avoid name conflicts between features in the model and the features inherited from OclAny all names on the features of OclAny start with ‘ocl’. Although theoretically there may still be name conflicts, they can be avoided. One can also use the pathname construct to explicitly refer to the OclAny properties.

Features of OclAny, the instance of OclAny is called object.

- **object = (object2 : OclAny) : Boolean**
  True if object is the same object as object2.

- **object <> (object2 : OclAny) : Boolean**
  True if object is a different object as object2.
  \[\text{post: result} = \text{not (object} = \text{object2)}\]

- **object.oclType** : OclType
  The type of the object

- **object.oclIsKindOf(type : OclType) : Boolean**
  True if type is a supertype (transitive) of the type of object.
post: result = type.allSuperTypes->includes(object.oclType) or type = object->oclType

object.oclIsTypeOf(type : OclType) : Boolean
    True if type is equal to the type of object.
    post: result = (object.oclType = type)

object.oclAsType(type : OclType) : type
    Results in object, but of known type type.
    Results in Undefined if the actual type of object is not type or one of its subtypes
    pre: object.oclIsKindOf(type)
    post: result = object
    post: result.oclIsKindOf(type)

7.1.3 OclExpression

Each OCL expression itself is an object in the context of OCL. The type of the expression is OclExpression. This type and its features are used to define the semantics of features that take an expression as one of their parameters: select, collect, forAll, etc.

An OclExpression includes the optional iterator variable and type and the optional accumulator variable and type.

Features of OclExpression, the instance of OclExpression is called expression.

evaluationType : OclType
    The type of the object that results from evaluating expression.

7.1.4 Real

The OCL type Real represents the mathematical concept of real. Note that Integer is a subclass of Real, so for each parameter of type Real, one can use an integer as the actual parameter.

Features of Real, the instance of Real is called r.

r = (r2 : Real) : Boolean
    True if r is equal to r2.

r + (r1 : Real) : Real
    The value of the addition of r and r1.

r - (r1 : Real) : Real
    The value of the subtraction of r1 from r.

r * (r1 : Real) : Real
    The value of the multiplication of r and r1.

r / (r1 : Real) : Real
    The value of r divided by r1.
\begin{verbatim}
r.abs : Real
The absolute value of r
post: if r < 0 then result = -r else result = r endif

r.floor : Integer
The largest integer which is less than or equal to r.
post: (result <= r) and (result + 1 > r)

r.max(r2 : Real) : Real
The maximum of r an r2.
post: if r >= r2 then result = r else result = r2 endif

r.min(r2 : Real) : Real
The minimum of r an r2.
post: if r <= r2 then result = r else result = r2 endif

r < (r2 : Real) : Boolean
True if r1 is less than r2.

r > (r2 : Real) : Boolean
True if r1 is greater than r2.
post: result = not (r <= r2)

r <= (r2 : Real) : Boolean
True if r1 is less than or equal to r2.
post: result = (r = r2) or (r < r2)

r >= (r2 : Real) : Boolean
True if r1 is greater than or equal to r2.
post: result = (r = r2) or (r > r2)

7.1.5 Integer

The OCL type Integer represents the mathematical concept of integer.

Features of Integer, the instance of Integer is called i.

i = (i2 : Integer) : Boolean
True if i is equal to i2.

i + (i2 : Integer) : Integer
The value of the addition of i and i2.

i + (r1 : Real) : Real
The value of the addition of i and r1.

i - (i2 : Integer) : Integer
The value of the subtraction of i2 from i.
\end{verbatim}
i - (r1 : Real) : Real
   The value of the subtraction of r1 from i.

i * (i2 : Integer) : Integer
   The value of the multiplication of i and i2.

i * (r1 : Real) : Real
   The value of the multiplication of i and r1.

i / (i2 : Integer) : Real
   The value of i divided by i2.

i / (r1 : Real) : Real
   The value of i divided by r1.

i.abs : Integer
   The absolute value of i
   post: if i < 0 then result = - i else result = i endif

i.div( i2 : Integer) : Integer
   The number of times that i2 fits completely within i.
   post: result * i2 <= i
   post: result * (i2 + 1) > i

i.mod( i2 : Integer) : Integer
   The result is i modulo i2.
   post: result = i - (i.div(i2) * i2)

i.max(i2 : Integer) : Integer
   The maximum of i an i2.
   post: if i >= i2 then result = i else result = i2 endif

i.min(i2 : Integer) : Integer
   The minimum of i an i2.
   post: if i <= i2 then result = i else result = i2 endif

7.1.6 String

The OCL type String represents ASCII strings.

Features of String, the instance of String is called string.

string = (string2 : String) : Boolean
   True if string and string2 contain the same characters, in the same order.

string.size : Integer
   The number of characters in string.

string.concat(string2 : String) : String
The concatenation of \textit{string} and \textit{string2}.
\begin{align*}
\text{post: } \text{result.size} &= \text{string.size} + \text{string2.size} \\
\text{post: } \text{result.substring}(1, \text{string.size}) &= \text{string} \\
\text{post: } \text{result.substring}(\text{string.size} + 1, \text{string2.size}) &= \text{string2}
\end{align*}

\textit{string.toUpper} : String

The value of \textit{string} with all lowercase characters converted to uppercase characters.
\begin{align*}
\text{post: } \text{result.size} &= \text{string.size}
\end{align*}

\textit{string.toLower} : String

The value of \textit{string} with all uppercase characters converted to lowercase characters.
\begin{align*}
\text{post: } \text{result.size} &= \text{string.size}
\end{align*}

\textit{string.substring(lower : Integer, upper : Integer)} : String

The sub-string of \textit{string} starting at character number \textit{lower}, up to and including character number \textit{upper}.

\subsection{7.1.7 Boolean}

The OCL type Boolean represents the common true/false values.

Features of Boolean, the instance of Boolean is called \textit{b}.

\begin{align*}
b = (b2 : \text{Boolean}) : \text{Boolean} \\
\text{Equal if } \text{b} \text{ is the same as } b2.
\end{align*}

\begin{align*}
b \text{ or } (b2 : \text{Boolean}) : \text{Boolean} \\
\text{True if either } \text{b} \text{ or } b2 \text{ is true.}
\end{align*}

\begin{align*}
b \text{ xor } (b2 : \text{Boolean}) : \text{Boolean} \\
\text{True if either } \text{b} \text{ or } b2 \text{ is true, but not both.} \\
\text{post: } (b \text{ or } b2) \text{ and not } (b = b2)
\end{align*}

\begin{align*}
b \text{ and } (b2 : \text{Boolean}) : \text{Boolean} \\
\text{True if both } b1 \text{ and } b2 \text{ are true.}
\end{align*}

\begin{align*}
\text{not } b : \text{Boolean} \\
\text{True if } b \text{ is false} \\
\text{post: } \text{if } b \text{ then result = false else result = true endif}
\end{align*}

\begin{align*}
b \text{ implies } (b2 : \text{Boolean}) : \text{Boolean} \\
\text{True if } b \text{ is false, or if } b \text{ is true and } b2 \text{ is true.} \\
\text{post: } (\text{not } b) \text{ or } (b \text{ and } b2)
\end{align*}

\begin{align*}
\text{if } b \text{ then } (\text{expression1 : OclExpression}) \\
\text{else } (\text{expression2 : OclExpression}) \text{ endif : expression1.evaluationType}
\end{align*}

If \textit{b} is true the result is the value of evaluating \textit{expression1} otherwise result is the value of evaluating \textit{expression2}.
7.1.8 Enumeration

The OCL type Enumeration represents the enumerations defined in an UML model.

Features of Enumeration, the instance of Enumeration is called *enumeration*.

\[
\text{enumeration} = (\text{enumeration2} : \text{Boolean}) : \text{Boolean}
\]

Equal if *enumeration* is the same as *enumeration2*.

\[
\text{enumeration} <> (\text{enumeration2} : \text{Boolean}) : \text{Boolean}
\]

Equal if *enumeration* is not the same as *enumeration2*.

**post**: \( \text{result} = \text{not} (\text{enumeration} = \text{enumeration2}) \)

7.2 Collection-Related Types

The following sections define the features on collections. I.e. these features are available on Set, Bag and Sequence. As defined in this section, each collection type is actually a template with one parameter. ‘T’ denotes the parameter. A real collection type is created by substituting a type for the T. So Set (Integer) and Bag (Person) are collection types.

7.2.1 Collection

Collection is the abstract supertype of all collection types in OCL. Each occurrence of an object in a collection is called an element. If an object occurs twice in a collection, there are two elements. In this section the operations on Collections are defined that have identical semantics for all collection subtypes. Some operations may be defined with the subtype as well, which means that there is an additional postcondition or a more specialized return value.

The definition of several common operations is different for each subtype. These operations are not mentioned in this section.

Features of Collection, the instance of Collection is called *collection*.

\[
\text{collection} \rightarrow \text{size} : \text{Integer}
\]

The number of elements in the collection *collection*

**post**: \( \text{result} = \text{collection} \rightarrow \text{iterate}(\text{elem}; \text{acc} : \text{Integer} = 0 | \text{acc} + 1) \)

\[
\text{collection} \rightarrow \text{includes} (\text{object} : \text{OclAny}) : \text{Boolean}
\]

True if *object* is an element of *collection*, false otherwise.

**post**: \( \text{result} = (\text{collection} \rightarrow \text{count}(\text{object}) > 0) \)

\[
\text{collection} \rightarrow \text{count}(\text{object} : \text{OclAny}) : \text{Integer}
\]

The number of times that *object* occurs in the collection *collection*

**post**: \( \text{result} = \text{collection} \rightarrow \text{iterate}(\text{elem}; \text{acc} : \text{Integer} = 0 | \text{if elem = object then acc + 1 else acc endif}) \)

\[
\text{collection} \rightarrow \text{includesAll}(\text{c2} : \text{Collection}(\text{T})) : \text{Boolean}
\]

Does *collection* contain all the elements of *c2*?

**post**: \( \text{result} = \text{c2} \rightarrow \text{forAll}(\text{elem} | \text{collection} \rightarrow \text{includes}(\text{elem})) \)
collection->isEmpty : Boolean
Is collection the empty collection?
post: result = ( collection->size = 0 )

collection->notEmpty : Boolean
Is collection not the empty collection?
post: result = ( collection->size <> 0 )

collection->sum : T
The addition of all elements in collection. Elements must be of a type supporting addition (Integer and Real)
post: result = collection->iterate( elem; acc : T = 0 | acc + elem )

collection->exists(expr : OclExpression) : Boolean
Results in true if expr evaluates to true for at least one element in collection.
post: result = collection->iterate(elem; acc : Boolean = false | acc or expr)

collection->forAll(expr : OclExpression) : Boolean
Results in true if expr evaluates to true for each element in collection. Otherwise result in false.
post: result = collection->iterate(elem; acc : Boolean = true | acc and expr)

collection->iterate(expr : OclExpression) : expr.evaluationType
Iterates over the collection. See section 6.5 for a complete description. This is the basic collection operation with which the other collection operations can be described.

7.2.2 Set

The Set is the mathematical set. It contains elements without duplicates.

Features of Set, the instance of Set is called set.

set->union(set2 : Set(T)) : Set(T)
The union of set and set2
post: T.allInstances->forAll(elem | result->includes(elem) = set->includes(elem) or set2->includes(elem))

set->union(bag : Bag(T)) : Bag(T)
The union of set and bag.
post: T.allInstances->forAll(elem | result->count(elem) = set->count(elem) + bag->count(elem))

set = (set2 : Set) : Boolean
Evaluates to true if set and set2 contain the same elements.
post: result = T.allInstances->forAll(elem | set->includes(elem) = set2->includes(elem))
set→intersection(set2 : Set(T)) : Set(T)
The intersection of set and set2. I.e. the set of all elements that are in both set and set2.
post: T.allInstances→forAll(elem | 
result→includes(elem) = 
set→includes(elem) and set2→includes(elem))

set→intersection(bag : Bag(T)) : Set(T)
The intersection of set and bag
post: result = set→intersection( bag→asSet )

set – (set2 : Set(T)) : Set(T)
The elements of set, which are not in set2
post: T.allInstances→forAll(elem | 
result→includes(elem) = 
set→includes(elem) and not set2→includes(elem))

set->including(object : T) : Set(T)
The set containing all elements of set plus object
post: T.allInstances→forAll(elem | 
result→includes(elem) = 
set→includes(elem) or (elem = object))

set->excluding(object : T) : Set(T)
The set containing all elements of set without object
post: T.allInstances→forAll(elem | 
result→includes(elem) = 
set→includes(elem) and not(elem = object))

set->symmetricDifference(set2 : Set(T)) : Set(T)
The sets containing all the elements that are in set or in set2, but not in both
post: T.allInstances→forAll(elem | 
result→includes(elem) = 
set→includes(elem) xor set2→includes(elem))

set->select(expr : OclExpression) : Set(expr.type)
The subset of set for which expr is true
post: result = set→iterate(elem; acc : Set(T) = Set{} | 
if expr then acc→including(elem) else acc endif)

set->reject(expr : OclExpression) : Set(expr.type)
The subset of set for which expr is false
post: result = set→select(not expr)

set->collect(expression : OclExpression) : Bag(expression.oclType)
The Bag of elements which results from applying expr to every member of set
post: result = set→iterate(elem; acc : Bag(T) = Bag{} | 
acc→including(expr) )

set->count(object : T) : Integer
The number of occurrences of object in set
post: result <= 1

set->asSequence : Sequence(T)
A Sequence that contains all the elements from \(set\), in random order.

\[
\text{post: } T\text{.allInstances->forAll(elem | result->count(elem) = set->count(elem))}
\]

\(set->\text{asBag} : \text{Bag}(T)\)

The Bag that contains all the elements from \(set\).

\[
\text{post: } T\text{.allInstances->forAll(elem | result->count(elem) = set->count(elem))}
\]

### 7.2.3 Bag

A bag is a collection with duplicates allowed. That is, one object can be an element of a bag many times. There is no ordering defined on the elements in a bag.

Features of Bag, the instance of Bag is called \(bag\).

\[
\text{bag} = (\text{bag2} : \text{Bag}) \iff \text{Boolean}
\]

True if \(bag\) and \(bag2\) contain the same elements, the same number of times.

\[
\text{post: result = T\text{.allInstances->forAll(elem | bag->count(elem) = bag2->count(elem))}}
\]

\(\text{bag->union(bag2} : \text{Bag}) : \text{Bag}(T)\)

The union of \(bag\) and \(bag2\)

\[
\text{post: T\text{.allInstances->forAll(elem | result->count(elem) = bag->count(elem) + bag2->count(elem))}}
\]

\(\text{bag->union(set} : \text{Set}) : \text{Bag}(T)\)

The union of \(bag\) and \(set\)

\[
\text{post: T\text{.allInstances->forAll(elem | result->count(elem) = bag->count(elem) + set->count(elem))}}
\]

\(\text{bag->intersection(bag2} : \text{Bag}) : \text{Bag}(T)\)

The intersection of \(bag\) and \(bag2\)

\[
\text{post: T\text{.allInstances->forAll(elem | result->count(elem) = bag->count(elem).min(bag2->count(elem))}})
\]

\(\text{bag->intersection(set} : \text{Set}) : \text{Set}(T)\)

The intersection of \(bag\) and \(set\)

\[
\text{post: T\text{.allInstances->forAll(elem | result->count(elem) = bag->count(elem).min(set->count(elem))}})
\]

\(\text{bag->including(object} : \text{T}) : \text{Bag}(T)\)

The bag containing all elements of \(bag\) plus \(object\)

\[
\text{post: T\text{.allInstances->forAll(elem | if elem = object then result->count(elem) = bag->count(elem) + 1 else result->count(elem) = bag->count(elem) endif)}}
\]
bag->excluding(object : T) : Bag(T)

The bag containing all elements of bag apart from all occurrences of object
post: T.allInstances->forAll(elem |
  if elem = object then
    result->count(elem) = 0
  else
    result->count(elem) = bag->count(elem)
endif)

bag->select(expression : OclExpression) : Bag(T)

The sub-bag of bag for which expression is true
post: result = bag->iterate(elem; acc : Bag(T) = Bag{} |
  if expr then acc->including(elem) else acc endif)

bag->reject(expression : OclExpression) : Bag(T)

The sub-bag of bag for which expression is false
post: result = bag->select(not expr)

bag->collect(expression: OclExpression) : Bag(expression.oclType)

The Bag of elements which results from applying expression to every member of bag
post: result = bag->iterate(elem; acc : Bag(T) = Bag{} |
  acc->including(expr) )

bag->count(object : T) : Integer

The number of occurrences of object in bag

bag->asSequence : Sequence(T)

A Sequence that contains all the elements from bag, in random order.
post: T.allInstances->forAll(elem |
  bag->count(elem) = result->count(elem))

bag->asSet : Set(T)

The Set containing all the elements from bag, with duplicates removed.
post: T.allInstances(elem |
  bag->includes(elem) = result->includes(elem))

7.2.4 Sequence

A sequence is a collection where the elements are ordered. An element may be part of a sequence more than once.

Features of Sequence(T), the instance of Sequence is called sequence.

sequence->count(object : T) : Integer

The number of occurrences of object in sequence

sequence = (sequence2 : Sequence(T)) : Boolean

True if sequence contains the same elements as sequence2 in the same order.
post: result = Sequence(1..sequence->size)->forAll(index : Integer |
  sequence->at(index) = sequence2->at(index))
and
sequence->size = sequence2->size

sequence->union (sequence2 : Sequence(T)) : Sequence(T)

The sequence consisting of all elements in sequence, followed by all elements in sequence2
post: result->size = sequence->size + sequence2->size
post: Sequence{1..sequence->size}->forAll(index : Integer | sequence->at(index) = result->at(index))
post: Sequence{1..sequence2->size}->forAll(index : Integer | sequence2->at(index) = result->at(index + sequence->size))

sequence->append (object: T) : Sequence(T)

The sequence of elements, consisting of all elements of sequence, followed by object.
post: result->size = sequence->size + 1
post: result->at(result->size) = object
post: Sequence{1..sequence->size}->forAll(index : Integer | result->at(index) = sequence->at(index))

sequence->prepend(object : T) : Sequence(T)

The sequence consisting of all elements in sequence, followed by object
post: result->size = sequence->size + 1
post: result->at(1) = object
post: Sequence{1..sequence->size}->forAll(index : Integer | sequence->at(index) = result->at(index + 1))

sequence->subSequence(lower : Integer, upper : Integer) : Sequence(T)

The sub-sequence of sequence starting at number lower, up to and including element number upper.
post: if sequence->size < upper then
result = Undefined
else
result->size = upper - lower + 1 and
Sequence{lower..upper}->forAll(index : Integer | result->at(index - lower + 1) = sequence->at(lower + index - 1))
endif

sequence->at(i : Integer) : T

The i-th element of sequence.
post: i <= 0 or sequence->size < i implies result = Undefined

sequence->first : T

The first element in sequence
post: result = sequence->at(1)

sequence->last : T

The last element in sequence
post: result = sequence->at(sequence->size)

sequence->including(object : T) : Sequence(T)

The sequence containing all elements of sequence plus object added as the last element
post: result = sequence.append(object)
sequence->excluding(object : T) : Sequence(T)

The sequence containing all elements of sequence apart from all occurrences of object.
The order of the remaining elements is not changed.
post: result->includes(object) = false
post: result->size = sequence->size - sequence->count(object)
post: result = sequence->iterate(elem; acc : Sequence(T) = Sequence{} |
      if elem = object then acc else acc->append(elem) endif )

sequence->select(expression : OclExpression) : Sequence(T)

The subsequence of sequence for which expression is true
post: result = sequence->iterate(elem; acc : Sequence(T) = Sequence{} |
      if expr then acc->including(elem) else acc endif)

sequence->reject(expression : OclExpression) : Sequence(T)

The subsequence of sequence for which expression is false
post: result = sequence->select(not expr)

sequence->collect(expression : OclExpression) :

Sequence(expression.oclType)

The Sequence of elements which results from applying expression to every member of sequence

sequence->iterate(expr : OclExpression) : expr.evaluationType

Iterates over the sequence. Iteration will be done from element at position 1 up until the
element at the last position following the order of the sequence.

sequence->asBag() : Bag(T)

The Bag containing all the elements from sequence, including duplicates
post: T.allInstances->forAll(elem | result->count(elem) = sequence->count(elem))

sequence->asSet() : Set(T)

The Set containing all the elements from sequence, with duplicated removed
post: T.allInstances->forAll(elem | result->includes(elem) = sequence->includes(elem))

8. GRAMMAR FOR OCL

This section describes the grammar for OCL expressions. An executable LL(1) version of this
grammar is available on the OCL web site mentioned at the start of this document.

The grammar description uses the EBNF syntax, where "|" means a choice , "?" optionality and
"*" means zero or more times. In the description of the name, typeName and string the syntax for
lexical tokens from the JavaCC parser generator is used. See http://www.suntest.com/JavaCC.

expression               := logicalExpression
ifExpression             := "if" expression
"then" expression
"else" expression
"endif"

logicalExpression         := relationalExpression
relationalExpression := relationalOperator relationalExpression |
additiveExpression :
  ( relationalOperator additiveExpression )?
additiveExpression := multiplicativeExpression |
  ( addOperator multiplicativeExpression )|
multiplicativeExpression := unaryExpression |
  ( multiplyOperator unaryExpression )|
uminaryExpression := postfixExpression |
  ( unaryOperator postfixExpression )|
postfixExpression := primaryExpression ( "." | "->" ) featureCall |
primaryExpression := literalCollection |
  literal timeExpression? qualifier? featureCallParameters? |
    ifExpression |
featureCallParameters := "(" ( declarator )? ( actualParameterList )? ")" |
literal := <STRING> | <number> | "#" <name> |
enumerationType := "enum" "{" "#" <name> ( "," "#" <name> ) }" |
simpleTypeSpecifier := pathTypeName |
  enumerationType |
literalCollection := collectionKind "{" expressionListOrRange? "}" |
expressionListOrRange := expression |
  ( "," expression )+ |
    ( ".." expression ) |
featureCall := pathName timeExpression? qualifiers? featureCallParameters? |
declarator := ":" <name> |
pathTypeName := pathTypeName ( "::" pathTypeName )* |
pathName := pathTypeName ( "::" pathTypeName )* |
timeExpression := @ <name> |
actualParameterList := expression ( "," expression )* |
logicalOperator := "and" | "or" | "xor" | "implies" |
collectionKind := "Set" | "Bag" | "Sequence" | "Collection" |
relationalOperator := "=" | ">" | ">=" | "<=" |
addOperator := "+" | "-" |
multiplyOperator := "*" | "/" |
unaryOperator := "not" |
typeName := "A"-"Z" ( "a"-"z" | "0"-"9" | "A"-"Z" | ")" )* |
nname := "A"-"Z" ( "a"-"z" | "0"-"9" | "A"-"Z" | ")" )* |
number := "0"-"9" |
string := "'" ( \" | \|^\" | \" | \" | \" | \" ) |
  ["a"-"z", "A"-"Z", "0"-"9", "\", "]", "]", "]", "\", "]") |
  ["0"-"7"] ( ["0"-"7"] )? |
  ["0"-"3"] ["0"-"7"] ["0"-"7"] |
string = "" ( \" | \|^\" | \" | \" | \" | \" ) |